

## Be a “Gyro”, Turn your SAS Report into a Sandwich

Michael Stout, Johnson & Johnson Medical Devices

### ABSTRACT

The title of this abstract may make you hungry. Bread is an essential part of any sandwich; just as titles and footnotes are an essential part of any report. Titles and footnotes provide context (the bread) and support for tables, figures and listings (TFLs). There are many methods for assigning titles and footnotes. This paper will show how standardization of titles and footnotes using a SAS® MACRO can improve the appearance and quality of deliverables. Key data points that should appear on every report will be reviewed. Lastly, the paper will present various techniques for controlling the format of titles and footnotes. Get ready for the best report sandwich ever!

### INTRODUCTION

Working in a fast-paced matrix organization has many challenges. Our team is comprised of statisticians and programmers from several device companies located across the country. The team does a lot of project related work that includes preparing reports for internal use, regulatory agencies, publications and investigator meetings. It is imperative that standards are implemented to improve overall efficiencies across the group. One area we focused on is the standardization of titles and footnotes.

Our goal was to have consistency and uniformity of titles and footnotes and be compliant with standard operating procedures (SOP) and work instructions. So far, we have looked at the reasons for standardizing titles and footnotes. Let’s begin looking at specific requirements and a sample table shell for titles and footnotes.

Here are the basic requirements:

1. Automation of selected titles and footnotes
2. Ability to assign values in titles and footnotes (i.e. system date and name of study)
3. Flexible and customizable solution
4. Compatible with Microsoft Word documents (RTF)
5. Easy to use

Sample Layout of standard titles and footnotes:

```
<Company Name> Confidential Extract Date: DDMMYYYY Status: DRAFT
Study: <Study Number> Page 1 of XXX
<Study description>

Program: xxxxxxxxxx, Run on <day> DDMMYYYY, HH:MM AM/PM
```

This time SAS® comes to the rescue to become the "Gyro"! Base SAS®, PROC REPORT®, ODS RTF® output destination, and the SAS MACRO® facility provide the tool set needed to implement a flexible solution that is easy to use. If you choose to read this paper you will learn how to automate creation of standard titles and footnotes and apply this concept to your specific business needs.

## MACRO %TITLES\_AND\_FOONOTES

You may be surprised at how simple and brief this code is. But what it does will amaze you. This macro can be easily modified to pass different values to the macro or to quickly change the layout of standard titles and footnotes. Named parameters are the primary method for passing values to the macro. Values returned from SAS functions, system macro variables (i.e. &SYSDATE) and user defined global variables may also be used to set values used by the macro. This macro uses these techniques to set values that effect used by the titles and footnotes.

The company name, study identifier, study name, extract date, status and notes used in the titles are passed to the macro using named parameters. &\_mypgm, which is used in Footnote1, is a user defined global macro variable that contains the name of the program. This macro variable must be set before calling the macro. Lastly, the date and time values used in Footnote1 are derived from SAS functions that return the current date and time.

The justification option on the title statement is used to place the text at the desired location on the report. Text may be centered, left justified or right justified. An inline modifier is used to control page numbering in the RTF output file. This macro was designed to work with output sent to the ODS RTF output destination, but could be modified to work with other ODS destinations too.

Macro to format standard titles and footnotes:

```
%MACRO Titles_and_Footnotes (Company = ,
                             StudyID = ,
                             StudyNm = ,
                             ExtractDt = ,
                             Status =,
                             Notes = Confidential);

* Get the current date, time (12-hour clock) and day of the week *;
%LOCAL curr_date curr_time curr_day;
%LET Curr_date = %SYSFUNC (LEFT (%SYSFUNC (DATE (), date9.)));
%LET Curr_time = %SYSFUNC (LEFT (%SYSFUNC (TIME (), timeAMP8.)));
%LET Curr_day = %SYSFUNC (LEFT (%SYSFUNC (DATE (), weekdate9.)));

TITLE1 j=1 "&Company "
       j=c "&Notes."
       j=r "Extract Date: &ExtractDT Status: &Status";
TITLE2 j=1 "Study: &StudyID"
       j=r "Page ^{Pageof}";
TITLE3 j=c "&StudyNm";

FOOTNOTE1 j=1 "Program: &_mypgm" ", Run on &Curr_day., &Curr_date., at
&Curr_time.";

%MEND;
```

## USING THE MACRO

Now that you have learned about the macro; it's time to use it. Remember the titles and footnotes are the bread of the sandwich. They provide context and support the report. The following code segment illustrates how to use the %Titles\_and\_Footnotes macro. The macro is responsible for setting Title1, Title2, Title3 and Footnote1. All other titles and footnotes must be set after calling the macro. Additional titles may be added starting with Title4 and footnotes starting with Footnote2.

By default, SAS Graph® procedures embed titles and footnotes within each graph, but this can be overridden on the ODS statement by using options nogtitle and nogfootnote.

Code that shows how to use the macro to format standard headings and footnotes:

```
OPTIONS orientation=portrait nodate nonumber;

/*****
| Set ODS options - required for inline formats
*****/
ODS escapechar "^";

%GLOBAL _mypgm;
%LET _mypgm = Cars.sas;

%Titles_and_Footnotes(Company    = PhamaSUG,
                      StudyID    = QT-03,
                      StudyNm    = Quick Tips Paper QT-03,
                      ExtractDt  = 01MAR2018,
                      Status     = DRAFT
                      );

ODS RTF FILE='C:\users\mstout2\Class.rtf';

TITLE4 'Table 1. Student Demographics where age > 14';

PROC REPORT DATA=sashelp.class(WHERE=(age > 14));
COLUMNS Name Sex Age Height Weight;
DEFINE Sex    / STYLE=[cellwidth=.75IN JUST=c];
DEFINE Age    / STYLE=[cellwidth=.75IN JUST=c];
DEFINE Height / STYLE=[cellwidth=.75IN] FORMAT=5.1;
DEFINE Weight / STYLE=[cellwidth=.75IN] FORMAT=5.1;
run;

ODS RTF close;
run;
```

## Output 1. Output from a PROC REPORT 1

<i>PhamaSUG</i>	<i>Confidential</i>	<i>Extract Date: 01MAR2018 Status:</i> <i>DRAFT</i>		
<i>Study: QT-03</i>		<i>Page 1 of 1</i>		
<i>Quick Tips Paper QT-03</i> <i>Table 1. Student Demographics where age &gt; 14</i>				
<b>Name</b>	<b>Sex</b>	<b>Age</b>	<b>Height</b>	<b>Weight</b>
Janet	F	15	62.5	112.5
Mary	F	15	66.5	112.0
Philip	M	16	72.0	150.0
Ronald	M	15	67.0	133.0
William	M	15	66.5	112.0
<i>Program: Cars.sas, Run on Friday, 09MAR2018, at 1:56 PM</i>				

## KEY DATA POINTS

The data collected and displayed in the titles and footnotes serve an important purpose. They provide context for the TFLs, improve traceability for internal and external audits, and are helpful when researching issues or discrepancies that are found in the TFLs. For example, the extract date can be used to determine why certain data points exist in the database but do not appear on TFL. What data points are important to your organization?

Summary of key data points to consider adding to standard titles and footnotes:

1. Extract Date                      Date when snapshot of the data was exported from the database
2. Company                            Name of the operating company or entity the report is for
3. Study Identifier                    Short name or unique identifier of the clinical study or project
4. Study Name                         Detailed description of the study or project
5. Program Name                      Name of program used to generate the TFL
6. Path                                 Path name of program
7. Status                                Status of TLF (DRAFT, FINAL, TEST, PROD)
8. Run Date\Time                     Date and Time TLF was generated
9. Notes                                Note indicating that TFLs are "Confidential" or "For Internal Use Only"  
Default value is "Confidential"

The placement of key data points on the report is not important but having key data points is important to your organization.

## FORMATTING TITLES AND FOOTNOTES

SAS provides several ways to format titles and footnotes. This section of the paper will examine how titles and footnotes can be formatted using various techniques and provide a brief discussion on how the key data points were assigned.

### APPLYING FORMATS

SAS options may be used to format titles and footnotes. Changing the page orientation to landscape can increase the area available to display titles and footnotes which may prevent wrapping of text. SAS options nodate and nonumber can be used to prevent SAS from displaying the system date and time in the upper right-hand corner of the TFL.

SAS Options statement:

```
OPTIONS orientation=landscape nodate nonumber;
```

Title and footnote options may be used to control the color, size, font and justification of standard titles and footnotes. Inline modifiers start with character "A" and may be used to add subscripts and underline, italicize, and bold text.

The compute block in PROC REPORT® may be used to attach report specific footnotes to the body of a report. This is an excellent way to display report level footnotes without changing standard footnotes, which appear at the bottom of the page. Notice how inline modifiers are used to superscript footnote references on each footnote and corresponding define statement.

Compute block with line statements to attach footnotes to body of report:

```
COMPUTE after;  
LINE @1 '^n Footnotes:';  
LINE @1 '^1 Height unit in inches';  
LINE @1 '^2 Weight unit in pounds';  
ENDCOMP;
```

Define statements used to link footnotes to height and weight column headers:

```
DEFINE Height / 'Height1' STYLE=[cellwidth=.75IN] FORMAT=5.1;  
DEFINE Weight / 'Weight2' STYLE=[cellwidth=.75IN] FORMAT=5.1;
```

Style templates may be used to format titles and footnotes. The easiest way to manage the overall look and feel of titles, footnotes and the body of report is to change the ODS default style template by using a pre-defined style template or by creating a custom style template. Style templates may be used to control colors, fonts, and border schemes. Over fifty style templates are included with SAS 9.4. Look online or submit the following code in display manger to get a list of available style templates: PROC TEMPLATE; PATH SASHELP.TMPLMST; LIST STYLES; RUN;

Sample ODS RTF statement using pre-defined Journal styles template:

```
ODS RTF FILE='&path\CLASS.RTF' STYLE=Styles.Journal;
```

Be a "Gyro", Turn your SAS Report into a Sandwich, continued

After applying the formats to the titles and footnotes, the report procedure was re-run. See the changes to the report in Output 2. Notice that the standard footnote was not impacted by the report specific footnotes for height and weight.

## Output 2. Output from a PROC REPORT after formatting titles and footnotes

PharmaSUG Study: QT-03	Confidential	Extract Date: 01MAR2018 Status: FINAL Page 1 of 1		
Quick Tips Paper				
Table 1. Student Demographics where age > 14				
<i>Name</i>	<i>Sex</i>	<i>Age</i>	<i>Height<sup>1</sup></i>	<i>Weight<sup>2</sup></i>
Janet	F	15	62.5	112.5
Mary	F	15	66.5	112.0
Philip	M	16	72.0	150.0
Ronald	M	15	67.0	133.0
William	M	15	66.5	112.0
Footnotes:				
<sup>1</sup> Height unit in inches				
<sup>2</sup> Weight unit in pounds				
Program: Cars.sas, Run on Friday, 09MAR2018, at 2:02 PM				

## ASSIGNING VALUES TO KEY DATA POINTS

Most key data points are passed to the macro using named parameters. For this paper, I chose to assign the program name to global macro variable named `&_mypgm`. To make the routine more robust, consider updating the macro to derive the program name directly from SAS. All date and time key data points are captured by the macro, so there is no need assign these data points.

## CONCLUSION

Now it's your turn to be a "Gyro"! You have learned the steps necessary to automate titles and footnotes for individual projects and company-wide. Use the `%Titles_and_Footnotes` macro, title options, inline formats and style templates to ensure consistency, control the appearance of deliverables and ensure adherence to SOPs and work instructions. I recommend creating a SETUP file for each project. You can use the SETUP file to allocate filerefs, librefs, SAS options and call `%Titles_and_Footnotes`. You can then include the SETUP file at the beginning of every program. Go and turn your reports into a sandwich and enjoy the results!

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Michael Stout  
Johnson & Johnson Medical Devices  
MStout2@its.jnj.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.